

NAG Library Routine Document

F07VEF (DTBTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07VEF (DTBTRS) solves a real triangular band system of linear equations with multiple right-hand sides, $AX = B$ or $A^T X = B$.

2 Specification

```
SUBROUTINE F07VEF (UPLO, TRANS, DIAG, N, KD, NRHS, AB, LDAB, B, LDB,      &
                  INFO)
INTEGER          N, KD, NRHS, LDAB, LDB, INFO
REAL (KIND=nag_wp) AB(LDAB,*), B(LDB,*)
CHARACTER(1)     UPLO, TRANS, DIAG
```

The routine may be called by its LAPACK name *dtbtrs*.

3 Description

F07VEF (DTBTRS) solves a real triangular band system of linear equations $AX = B$ or $A^T X = B$.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

5 Arguments

- | | |
|--|--------------|
| 1: UPLO – CHARACTER(1) | <i>Input</i> |
| <i>On entry:</i> specifies whether A is upper or lower triangular.
UPLO = 'U'
A is upper triangular.
UPLO = 'L'
A is lower triangular.
<i>Constraint:</i> UPLO = 'U' or 'L'. | |
| 2: TRANS – CHARACTER(1) | <i>Input</i> |
| <i>On entry:</i> indicates the form of the equations.
TRANS = 'N'
The equations are of the form $AX = B$.
TRANS = 'T' or 'C'
The equations are of the form $A^T X = B$.
<i>Constraint:</i> TRANS = 'N', 'T' or 'C'. | |

3: DIAG – CHARACTER(1) *Input*

On entry: indicates whether A is a nonunit or unit triangular matrix.

DIAG = 'N'

A is a nonunit triangular matrix.

DIAG = 'U'

A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

Constraint: DIAG = 'N' or 'U'.

4: N – INTEGER *Input*

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

5: KD – INTEGER *Input*

On entry: k_d , the number of superdiagonals of the matrix A if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.

Constraint: $KD \geq 0$.

6: NRHS – INTEGER *Input*

On entry: r , the number of right-hand sides.

Constraint: $NRHS \geq 0$.

7: AB(LDAB,*) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array AB must be at least max(1, N).

On entry: the n by n triangular band matrix A .

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in AB($k_d + 1 + i - j, j$) for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in AB($1 + i - j, j$) for $j \leq i \leq \min(n, j + k_d)$.

If DIAG = 'U', the diagonal elements of A are assumed to be 1, and are not referenced.

8: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07VEF (DTBTRS) is called.

Constraint: $LDAB \geq KD + 1$.

9: B(LDB,*) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least max(1, NRHS).

On entry: the n by r right-hand side matrix B .

On exit: the n by r solution matrix X .

10: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07VEF (DTBTRS) is called.

Constraint: $LDB \geq \max(1, N)$.

11: INFO – INTEGER

Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle \text{value} \rangle$ of the diagonal is exactly zero. A is singular and the solution has not been computed.

7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham (1989).

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

$$|E| \leq c(k)\epsilon|A|,$$

$c(k)$ is a modest linear function of k , and ϵ is the **machine precision**.

If \hat{x} is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|\hat{x} - x\|_\infty}{\|\hat{x}\|_\infty} \leq c(k) \operatorname{cond}(A, x)\epsilon, \quad \text{provided } c(k) \operatorname{cond}(A, x)\epsilon < 1,$$

where $\operatorname{cond}(A, x) = \|A^{-1}\| |A| \|x\|_\infty / \|x\|_\infty$.

Note that $\operatorname{cond}(A, x) \leq \operatorname{cond}(A) = \|A^{-1}\| |A| \|_\infty \leq \kappa_\infty(A)$; $\operatorname{cond}(A, x)$ can be much smaller than $\operatorname{cond}(A)$ and it is also possible for $\operatorname{cond}(A^T)$ to be much larger (or smaller) than $\operatorname{cond}(A)$.

Forward and backward error bounds can be computed by calling F07VHF (DTBRFS), and an estimate for $\kappa_\infty(A)$ can be obtained by calling F07VGF (DTBCON) with NORM = 'T'.

8 Parallelism and Performance

F07VEF (DTBTRS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07VEF (DTBTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $2nkr$ if $k \ll n$.

The complex analogue of this routine is F07VSF (ZTBTRS).

10 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} -4.16 & 0.00 & 0.00 & 0.00 \\ -2.25 & 4.78 & 0.00 & 0.00 \\ 0.00 & 5.86 & 6.32 & 0.00 \\ 0.00 & 0.00 & -4.82 & 0.16 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -16.64 & -4.16 \\ -13.78 & -16.59 \\ 13.10 & -4.94 \\ -14.14 & -9.96 \end{pmatrix}.$$

Here A is treated as a lower triangular band matrix with one subdiagonal.

10.1 Program Text

```
Program f07vefe

!     F07VEF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
Use nag_library, Only: dtbtrs, nag_wp, x04caf
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
Character (1), Parameter :: diag = 'N', trans = 'N'
!     .. Local Scalars ..
Integer :: i, ifail, info, j, kd, ldab, ldb, n, &
           nrhs
Character (1) :: uplo
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ab(:,:), b(:,:)
!     .. Intrinsic Procedures ..
Intrinsic :: max, min
!     .. Executable Statements ..
Write (nout,*) 'F07VEF Example Program Results'
!     Skip heading in data file
Read (nin,*)
Read (nin,*) n, kd, nrhs
ldb = kd + 1
ldb = n
Allocate (ab(ldb,n),b(ldb,nrhs))

!     Read A and B from data file

Read (nin,*) uplo
If (uplo=='U') Then
    Do i = 1, n
        Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
    End Do
Else If (uplo=='L') Then
    Do i = 1, n
        Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
    End Do
End If
Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Compute solution
!     The NAG name equivalent of dtbtrs is f07vef
Call dtbtrs(uplo,trans,diag,n,kd,nrhs,ab,ldab,b,ldb,info)

!     Print solution

Write (nout,*)
Flush (nout)
If (info==0) Then
    !     ifail: behaviour on error exit
          !     =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
!
```

```

ifail = 0
Call x04caf('General', ' ', n, nrhs, b, ldb, 'Solution(s)', ifail)

Else
  Write (nout,*) 'A is singular'
End If

End Program f07vefe

```

10.2 Program Data

```

F07VEF Example Program Data
 4   1   2                      :Values of N, KD and NRHS
 'L'                           :Value of UPLO
 -4.16
 -2.25   4.78
          5.86   6.32
          -4.82   0.16   :End of matrix A
-16.64   -4.16
-13.78   -16.59
 13.10   -4.94
-14.14   -9.96                 :End of matrix B

```

10.3 Program Results

F07VEF Example Program Results

Solution(s)		
	1	2
1	4.0000	1.0000
2	-1.0000	-3.0000
3	3.0000	2.0000
4	2.0000	-2.0000
